
Abstract Thehistory, benefits,anddrawbacksto purerule-
basedprogrammingis discussed.A simpleextensionto pure
rule-basedprogrammingisdescribed.Theextensionsarevery
quick to code and can be easily customizedto supportsa
rangeof knowledgeengineeringapplications.

1 Intr oduction

At a workshopon rule-basedprogramming(hereafter, RBP),
it maybeheresyto saythat thereis moreto knowledgethan
just rules.However, aftermany yearsof commercialandre-
searchwork on RBP, we assertthatthis is so.

This article reviews someof the history of RBP andthe
needto extendcertainaspectsof RBP. Theseextensionsare
simpleto implement-sosimplein fact that theentiresource
codefor thoseextensionscanbepresentedin this article.

2 A DummiesGuide to RBP

2.1 Origins & Early Successes

This article focuseson rule-basedknowledgeengineering.
Hence,by “RBP”, we really mean“how ruleswereusedin
classicalknowledgeengineering”.

Much of the early 1980shype surroundingcommercial
applicationsof artificial intelligencecamefrom early suc-
cesseswith rule-basedproductionsystems. Suchsystemswere
rule-basedsystemsthatqueriedandupdateobjectsin awork-
ing memoryusingaMATCH-SELECT-ACT cycle:

– MATCH: find the ruleswith conditionssatisfiedby the
currentcontentsof workingmemory;

– SELECT: pick onerule from the MATCHedsetusinga
conflictresolutionstrategy;

– ACT: performtheactionof thepickedrule.

Thereare many advantagesto pure RBP. For example,the
uniformity of theRPGparadigmmakesit amenableto:

– formalanalysisof their reliability, e.g.[5];
– powerful learningschemeslanguages,e.g.[8];
– the rapidcreationof high-productivity programmingen-

vironments,e.g.[7,11,12];
– therapidtrainingof businessuserssothatthey cancreate

their own rule bases,e.g.[16,17];
– powerful maintenanceenvironments,e.g.[6,19].

Further, RBPis aninsightful theoreticaltool for cognitive
psychology. PureRBPcanreplicatecertainexpertanderro-
neousbehavior of experts.For example,oneway to explain
thedifferencebetweenexpertandnoviceperformanceis that
novicesfill their working memorywith an excessof active
goals.This leavesnoroomfor any intermediariesof any par-
ticular calculation.On theotherhand,expertshavecompiled
their experienceinto high-priority rules that selectthe next
bestaction.Hence,theworkingmemoryof anexperthasless
activegoalswhichmeansexpertsarefreeto usetheirmemory
to runcomputations[9].

Not only is RBPusefulfor cognitive theory, it is a useful
tool for pragmaticsoftwareengineers.RBP enablesa novel
iterativeandexploratorysoftwaredevelopmentmethodology.
Iterativeandexploratorysoftwaredevelopmentis veryuseful
whenprototypingsoftware.Suchprototypingis not required
for well-definedtasks.Suchwell-definedtaskscanbeimple-
mentedvia a “waterfall” developmentprocess;i.e.

For less-definedtasks,waterfall developmentcan stag-
natein the analysisstagesincenot enoughis known about
thedomain.An alternative approachis to useRBPto gener-
ateanexecutableversionof thecurrentconceptualizationof
a system.Sinceeachrule is a separatechunkof knowledge,
it is easyto quickly addmorerules.This rule basewill, most
probably, beincomplete.However, on execution,theinterac-
tion of theserulescanleadto surprisingresultsthatprompt
clarificationsandextensionsof domainknowledge.This ap-
proachhasbeencalledvariousthingsincluding“knowledge
elicitationvia irritation” or theRUDE model;i.e.

+-,-.0/ � + 1 �2� , �� �	��3�4�
���� 5� . 	�6 1 #7� /  8�9�



RBP methodsresultedin the “AI spring” of the 1980s.
Many well-documented,mature,andoptimizedRBPsystems
weredevelopedsuchasART 1, CLIPS2, andOPS5[3] (justto
namea few). NumeroussignificantRBPsystemsweredevel-
opedincludingthecommerciallysuccessfulXCON computer
configurationsystem[13].

2.2 Problemswith Rules

The blossomingof RBP in the AI springwasnot followed
by an RBP summer. An assumptionunderlyingthe RUDE
approachwas that rules are independentchunksof knowl-
edgewhichcanbeeasilyaddedor changedor removed.This
provednot to bethecase.For example,onceXCON grew to
10,000rules,thedevelopersof XCON hadaRUDE3 awaken-
ing: maintainingXCON’s ruleshadbecomefiendishlycom-
plicated.Tosomeextent,thiswasdueto thedensityof knowl-
edgewithin XCON. The expertisewithin XCON’s rulesre-
flectedDEC’sstate-of-the-artknowledgein configuringcom-
putersandsucha rich library of knowledgewill be intricate
to maintain,no matterhow it is expressed.However, another
factorcomplicatedXCON’s maintenancewasthat it’s rules
violatedthe RUDE assumption.Real-world rule basesoften
containedgroupsof rules with significant interactions.For
example:

– A carefulreverseengineeringof XCON showedthat the
systemexecutedthoughseveral operator spaceswhere
methodsfor improving the designof a computerwere
carefully collected,rejectedor elaborated,assessed,be-
foretheappropriatebestoperatorwasfinally selected[1].

– Figure1 shows threerulesthat checkfor certainspecial
casesof bagginggroceries.Theserulescan’t be under-
stoodin isolationwith theothers.Ruleb11 triesto sneak
small itemsinto grocerybagsthat aren’t full andwhich
don’t containbottles.If b11fails,thenruleb12justplaces
small itemsinto any grocerybagat all. Ruleb13creates
a new badwhenneitherof theothertwo rulescanfind a
bagfor smallitems.Notethetacit relianceof b12on b11
handlinga certainspecialcase(bagswith bottles).Note
alsothetacit relianceof b13ontheothertwo rules:creat-
ing emptygrocerybagsis a nonsenseactionunlesssome
otheragenttriesto first fill thosebags.

SuchcoordinatingrulesviolatetheRUDEassumptionsince
everyadditionto therulebasehasto beassessedwith respect
to its effecton therestof therules.

Anotherproblemwith pureRBPis thattheparadigmcan
confuse,not clarify, certaintypesof proceduralknowledge.
For example,the processof finding the volumesof differ-
ent itemsin a grocerybag.Onegenerator rule is requiredof

1 FromInferenceCorporation
2 The“C” LanguageIntegratedProductionSystem,developedby

NASA [18]
3 Pun.Function:noun.Etymology:perhapsfrom Italianpuntiglio

finepoint,quibble.Definition:theusuallyhumoroususeof awordin
suchawayasto suggesttwo or moreof its meaningsor themeaning
of anotherwordsimilar in sound.Sorry.

smallitems.pl

b11
in bag_small_items
if order=I with ‘items has N and

grocery with ‘name=N with ‘size=small and
bag=B with *notFull and
not (bag=B with ‘contents has C and

grocery with ‘name=C with *type(bottle))
then change order=I with delete(‘items,N,!items) and

change bag=B with ‘contents takes N
since ’best to avoid bottles and small items’.

b12
in bag_small_items
if order=I with ‘items has N and

grocery with ‘name=N with ‘size=small and
bag=B with *notFull

then change order=I with delete(‘items,N,!items) and
change bag=B with ‘contents takes N

since ’sneaking a small item into a not full bag’.

newbag4small
in bag_small_items
if order with ‘items has N and

grocery with ‘name=N with ‘size=small
then make bag with *nothing
since ’need a new bag’.

Fig. 1 Three rules in the PIKE language(a STARLOG variant).
Theserulesaccesstheobjectdefinedin Figure2, Figure3, andFig-
ure4. Exampleadaptedfrom Winston’s BAGGERapplication[23].

grocery.pl

groceryDB(1, bread, bag(plastic), medium, n).
groceryDB(2, glop, jar, small, n).
groceryDB(3, granola, box(cardboard), large, n).
groceryDB(4, iceCream, carton(cardboard), medium, y).
groceryDB(5, pepsi, bottle, large, n).
groceryDB(6, potatoChips, bag(plastic), medium, n).

% GROCERYhas five fields, none of which are indexed
grocery=groceryDB(id,name,type,size,frozen).

% define GROCERYtypes
grocery*type(T) --> functor(‘type,T,_). % : . . . . . . . 12

% size symbols to numbers
grocery*volumes([small/1, medium/2, large/3]).

% accessing the numeric size of a particular size symbol
grocery*volume(V) --> * volumes(Vs), ‘size = S, Vs has S/V.

Fig. 2 A PIKE definitionof theGROCERY object.

order.pl

% ORDERhas two fields and the first one is indexed
order=orderDB(+id, % "+" denotes indexing

items ).

order*size(20). % max number of items in an order
order*active. % ORDERis "active"; i.e. delete all at reset

% Accessing the GROCERYterm with a certain Name.
% GROCERYdefined in Figure 2
order*item(Name,X) --> grocery with ‘name=Name with ‘self=X.

% backtracks through all GROCERYitems that are items
% in this ORDER
order*item(Item) --> ‘items has Name, *item(Name,Item).

Fig. 3 A PIKE definitionof theORDERobject.

bag.pl

bag=bagDB(+id,contents).

bag*active.
bag*capacity(20).
bag*empty --> ‘contents=[].

bag*newBag(Id,Contents) --> flag(ids,Id,Id+1),
!id=Id, %: . . . . . . . . . . . . . . . . 8
!contents=Contents.

bag*nothing --> *newBag(_,[]).

bag*largeItem(I) --> grocery with ‘name=I with ‘size=large.

bag*largeItems1(Item,1) --> *largeItem(Item),!.
bag*largeItems1(_,0).

bag*largeItems([H|T],N0+N) --> *largeItems1(H,N0),
*largeItems(T,N).

bag*largeItems([],0).

bag*largeItems(N) --> ‘contents=Items,
*largeItems(Items,N0),
N is N0.

bag*volume([Item|Items],V0+V) --> %: . . . 25
grocery with ‘name=Item with *volume(V0),
*volume(Items,V).

bag*volume([],0). %: . . . 28

bag*volume(V) -->
‘contents=Items, *volume(Items,V0), V is V0.

bag*full --> * volume(V), *capacity(S), V >= S. %: 33

bag*notFull --> not (* full).

Fig. 4 A PIKE definitionof theBAG object.

transferringpairsof groceryitemsfrom that setto a tempo-
rary spaceof “candidatesums”.Another intermediaryrule
matchesanddeleteseachpair, thenassertsthe sumof their
sumasanothermemberof the“candidatesums”.A final re-
port rule waits till the generatorand intermediaryrule stop
firing, thenaccessesthesurviving “candidatesum”astheto-
tal volumeof thegrocerybag.A moresuccinctrepresentation
of this proceduralsummationknowledgeis the list summa-
tion procedureshown in Figure4 betweenline 25andline 28.

Many otherresearchersarguedthatruleswerenot theap-
propriateprimitive constructof knowledgeengineering.De-
spitecarefulattemptsto generalizethe early knowledgeen-
gineeringwork (e.g. [22]), the constructionof knowledge-
basedsystemsremaineda somewhat hit-and-missprocess.
By the endof the 1980s,it wasrecognizedthatdesigncon-
ceptssuchas RBP were incomplete[4]. For example,Bo-
brow’s reverseengineeringof real-world knowledge-based
systems[2] found thatnumerousparadigmswerebeingem-
ployed includingrule-based,logic-based,functional,object-
oriented,and “access-based”(which, thesedays,we might
call implicit invocation[21]). The 1990swas characterized
byanextensivesearchfor higher-level reusablepatternsof in-
ferencesuchaspropose-and-revise(e.g.asdoneby [20] or a
recursivedescentof “problemspaces”(e.g.asdoneby [24]).

business logic
(rules & objects)

Fig. 5 The“iceberg model” of knowledgeengineering.

2.3 BeyondRBP

Theaboveproblems,andourown commercialknowledgeen-
gineering(e.g. [16,17]), leadus to extent RBP. Like many
others(e.g. ART, CLIPS), the needto useboth procedural
anddeclarativeruleknowledgemadeuscombineRBPwith a
simpleobject-orientedapproach.Ruleconditionsandactions
coulduseverbsdefinedin theobject-language.For example,
rule b12 in Figure1 checksthata bagis notFull andnotFull
is a proceduraldefinedat theendof bag in Figure4.

Also, for awhile,wetriedcodingupknowledgeengineer-
ing languagesbasedon thesupposedlyreusablehigher-level
reusablepatternsof inference.However, therewasaproblem.
Our repeatedexperiencewas that while small communities
of expertsmight reusean inferencepattern,thatpatternwas
not widely endorsedelsewhere.That is, while designinga
rule-basedaroundacertaininferencepatternwasuseful,each
new applicationneededanew inferencepattern(aneffect re-
portedelsewhere[10]). More generally, while many higher-
level inferencepatternshave beenidentified (e.g. propose-
and-revise,heuristicclassification,recursivedescentof “prob-
lem spaces’),thereusabilityof thesepatternsis questionable
sincethereneverwaswidespreadandstableagreementof the
internalstructureof thesepatterns[14,15].

Even thoughinferencepatternsmay not be reusablebe-
tweendomains,they may be useful within a particulardo-
main. Our default architecturefor a new knowledgebased
systemwasthe iceberg modelof Figure5. In that architec-
ture,knowledgeengineerswork “underthewaterline”tobuild
infrastructureto supportthe“in view” knowledgebasescre-
atedby advancedbusinessusers.Our role asknowledgeen-
gineerswasto:

– Identify clichesin theexpert’sapproachto differentprob-
lems.

– Craft supportcodefor eachsuchcliche.

Wherepossible,thesupportcodewasheavily parameterized
sothat it couldbeextensively customized.Thesecustomiza-
tion parametersthenbecamethe“tip of theiceberg” thatwas
in view to our businessusers.Theseusersthen usedthese
upper-mostdriversin their rulesandobjects.Our clichesin-

Fig. 6 Patternsof rule growth. KE= knowledgeengineers

cludedlow-level idiomssuchassummingall itemsis a list as
well asdomain-specifichigh-level inferencepatterns.

Figure 6 shows a typical patternof authoringrules us-
ing this iceberg approach.Notethattheknowledgeengineers
write somerulesin theinitial stageswhile, by theendof the
development,the usershave written mostof the rules.This
patternof rule authoringarisesfrom the from the following
developmentmethodology:

Languagedevelopment:Initially, theknowledgeengineer(s)
struggleto understandthe domainandidentify the rele-
vant cliches.After a weekor two, someof thesecliches
arefoundandimplementedassupportcode.

Transition: Theknowledgeengineerthenbuilds a few sam-
ple rules to demonstratethe usageof the supportcode.
Thesesamplerules are then usedto train the business
users.

Development:Businessusersgoontowrite mostof theknowl-
edgebase.

Languageelaboration: Theknowledgeengineerwatchestheir
progressto identify commoninferencecliches that are
awkwardor clumsyor errorproneto encode.Theknowl-
edgeengineersthen (i) augmentthe supportcode and
(ii) show the businessusershow to simplify their rules
usingthe augmentedsupportcode.As a result,thebusi-
nessuserslearnhow to encodeandupdatetheir own rule
baseusinga knowledgelanguagethat hasbeenheavily
customizedto their domain.

Maintenance:Maintenancein thisapproachis relativelysim-
ple sincebusinessuserscanupdatetheir own knowledge
evenwhentheknowledgeengineeris unavailable.

3 STARLOG

Theiceberg modelis only possiblewhenthepractitionercan
quickly craft a new setof inferencecliches.The restof this
paperdiscussesSTARLOG,acustomizationtool kit for knowl-
edgeengineering.

STARLOG systemis a setof load-timemacros thatcon-
vert sentencesin somedomain-specificterminology into a
simpleclause-basedlogic. Sincethesemacrosarecalledat

ops.pl

:- op(999, xfx , if).
:- op(998, xfx , then).
:- op(997, xfx, since).
:- op(996, fx, say).
:- op(990, xfy , or).
:- op(989, xfy , and).
:- op(988, fy , not).
:- op(980, fx, [make,change,zap]).
:- op(970, xfy , with).
:- op(969, xfx , takes).
:- op(968, xfx , has).
:- op(1, fx , goto).
:- op(1, xfy , [at,in]).
:- op(1, fx , [‘,!,*]).

Fig. 7

starlog.pl

:- [flags]. % see Figure 23

@ X :- (option(loadSlowly)
-> Options= []
; Options=[silent(true), if(changed)]),

load_files(X,Options).

:- @[%%%standard start up files
ops % see Figure 7

,hooks % see Figure 9
,hacks % see Figure 27

%%%some general library routines
,show % see Figure 29
,tidy % see Figure 18
,demos % see Figure 30
,singleton % see Figure 28
,misc % see Figure 31
,sharedvars % see Figure 17

%%%code specific to rules and objects in Prolog
% the object system:
,spec % see Figure 12, & Figure 10
,wrapper % see Figure 24
,ecg % see Figure 13
,verbs % see Figure 25
% the rule system:
,rules % see Figure 15, Figure 19 & Figure 22
,fchain % see Figure 16
%,egs % see Figure 26 (uncomment to see demos)
].

Fig. 8 Theidiom @[File1, File2,..] is shorthandfor “don’t
loadthesefilesmorethatonceunlessthey have notchangedondisc
and,if loading,don’t print verboseloadmessages”.

loadtimethen,in many cases,theoverheadsof interpretation
is incurredonceat loadtimeandneverat runtime.

STARLOG is aProlog-basedframework for building dif-
ferentlanguagesfor knowledgeengineering.Prologwascho-
senastheunderlyingimplementationlanguagedueto its ease
of customization.For example,Figure7 changestheProlog
interpretersuchthat certainuser-friendly structurescan be
createdin apseudo-Englishstyle;e.g.therulesshown in Fig-
ure1.

Thispaperpresents367linesof Prologthatimplementsa
simple,but usefulrule/objectinterpreter/optimizer. Themain
loadfile of STARLOG is shown in Figure8.Of thiscode,127
lines aresupportutilities; 153 implementan optimizedsim-
ple objectlanguageand87 linesimplementa forwardchain-
ing rule-basedsystem.Sucha smallsetof utilities caneasily

hooks.pl

% OBJECTS:
term_expansion(Helper=Spec, X) :-

spec(Helper=Spec,X).

% METHODS:
term_expansion((Helper*Head --> Body), X) :-

ecg((Helper*Head --> Body),X).
term_expansion(Helper*Head, X) :-

ecg((Helper*Head --> true),X).

% RULES:
term_expansion(Label if Condition then Action,X) :-

rules(Label if Condition then Action,X).

Fig. 9

be customizedfor new domains.Onesuchcustomizationis
the PIKE language4 usedfor the rulesandobjectsshown in
Figure1, Figure2, Figure3 andFigure4.

PIKE supportsthreemainconstructsshown in Figure9:
objects,methods,andrules.Theseconstructsarediscussed
below Dueto spacerestrictions,this discussionwill bequite
terse.A longer versionof this article is underpreparation
which explainsthesystemin moredetail.

3.1 Objects

Knowledgebaseauthors(hereafter, authors)candefineob-
jectsusingtheidiom Helper=Spec whereSpec describes
thestructureof aworkingmemoryobject. EachPIKE object
containsa setof namedfields,someof which aremarkedby
a+ if they areindex fields.GROCERY itemshavenoindexes
(seeFigure2) but ORDERsandBAGshaveoneindexedfield
each(seeFigure3 andFigure4). Helper is the nameof a
predicatethatcanaccesspartsof anobject.For example,the
Figure 10 shows the internal Prolog representationof Fig-
ure 4. For example,the grocery/5 predicatesstartingat
line 14 allow the accessandupdateof fieldswithin a GRO-
CERY object.

The Helper=Spec idiom arehandledby the spec/2
predicatedefinedin Figure12.Currently, PIKE’sobjectssup-
portencapsulationandpolymorphism,but not inheritance.

3.2 Methods

The idiom Helper*Head --> Body is PIKE’s method
syntax.Thesemethodsareanextensionto thestandardPro-
log definiteclausegrammarandso arecalledECGs. ECGs
cancontainnamedreferencesto objects.Within ECGs,the
idiom ‘X accessesthecurrentvalueof field X andthe idiom
!X updatesfield X. For example:

4 For StarTrek aficionados,we offer thefollowing notes.STAR-
LOG variantsshouldbe named,in order, after the captainsof the
Rodenberry-classstar ships: i.e. PIKE KIRK, SPOCK,PICARD,
SISKO,JANEWAY, DA’AN andARCHER.ThenamesSTOCKER,
DECKER, KAHN andZO’OR arereserved for throw-away crazy
prototypes,for obviousreasons.

speceg.out

% output from speceg.pl

grocery(A) :-
grocery(A, B, C).

grocery(type(A), groceryDB(B,C,D,E,F), % : . . . . . . . . . . . 6
groceryDB(B,C,D,E,F)) :-

functor(D,A,G). % : . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 8
grocery(volumes([small/1, medium/2, large/3]), A, A).
grocery(volume(A), groceryDB(B, C, D, E, F),

groceryDB(B, C, D, E, F)) :-
[small/1, medium/2, large/3]has E/A.

grocery(self, A, B, A, B). % : . . . . . . . . . . . . . . . . . . . . . . . . 14
grocery(id, A, B, groceryDB(A, C, D, E, F),

groceryDB(B, C, D, E, F)).
grocery(name, A, B, groceryDB(C, A, D, E, F),

groceryDB(C, B, D, E, F)).
grocery(type, A, B, groceryDB(C, D, A, E, F),

groceryDB(C, D, B, E, F)).
grocery(size, A, B, groceryDB(C, D, E, A, F),

groceryDB(C, D, E, B, F)).
grocery(frozen, A, B, groceryDB(C, D, E, F, A),

groceryDB(C, D, E, F, B)).

’:spec’(grocery, groceryDB, groceryDB/5
[], [0, 0, 0, 0, 0], [id, name, type, size, frozen]).

:-index(groceryDB/5).
:-dynamic groceryDB/5.

portray(groceryDB(B, C, D, E, F)) :-
write(groceryDB/5).

touch(groceryDB(A, B, C, D, E), F, G) :-
grocery(F, groceryDB(A, B, C, D, E), G).

gets(grocery, groceryDB(B, C, D, E, F), []) :-
groceryDB(B, C, D, E, F).

sets(grocery, groceryDB(A, B, C, D, E), []) :- % : 41
bretract(groceryDB(B, C, D, E, F)),
bassert(groceryDB(A, B, C, D, E)).

makes(grocery, groceryDB(B, C, D, E, F), []) :- % : 45
bassert(groceryDB(B, C, D, E, F)).

zaps(grocery, groceryDB(B, C, D, E, F), []) :-
bretract(groceryDB(B, C, D, E, F)).

goal_expansion(grocery(A, B, C), D) :- % : . . . . . . . . . 51
singleton(grocery(A, B, C)),
clause(grocery(A, B, C), D).

goal_expansion(grocery(B, C, D, D), grocery(B, C, C, D, D)).
goal_expansion(grocery(A, B, C, D, E), true) :-

one(grocery(A, B, C, D, E)).

Fig. 10 Outputfrom Figure11.

– Line 12of Figure2hasthecodefunctor(‘type,T, -
) . STARLOG’s load-timemacrosexpandthis into theac-
cessto the third field of groceryDB/5 (i.e. the type
field) shown in Line 8 of Figure10.

– Line 8 of Figure 4 has the code !id=Id . This idiom
forcesan updateof the first field of a bagDB/2 fact to
avaluecomputedfrom the flag/3 predicate5; i.e.:

bag(newBag(Id, Contents), _, bagDB(Id, Contents)) :- flag(ids,Id,Id+1)

TheHelper*Head --> Body idiom ishandledbyecg/2
definedin Figure13.

5 SWI Prolog’s flag/3 predicatemaintainsand updatesnu-
mericcounters.

speceg.pl

:- [starlog]. % see Figure 8

:- [grocery].

portray(’:spec’(A, B, C, D, E, F)) :-
format(’’’:spec’’(˜p, ˜p, ˜p’,[A,B,C]),nl,
format(’ ˜p, ˜p, ˜p).’,[D,E,F]).

speceg :-
listing(grocery),
spec(grocery=groceryDB(id,name,type,size,frozen ), List0),
none(List0,grocery,List),
show(List).

none([],_,[]).
none([(H:-_)|T],F, Rest) :- functor(H,F,_),!,none(T,F,Rest).
none([H|T], F, Rest) :- functor(H,F,_),!,none(T,F,Rest).
none([H|T], F,[H|Rest]) :- none(T,F,Rest).

:- demos(speceg).

Fig. 11 Starlogsample;generatesFigure10.

3.3 Forward ChainingRules

TheidiomLabel if Condition then Action isPIKE’s
wayof definingforwardchainingrules.Ruleshavepriorities
andgroupswhich canbespecifiedwithin theLabelThede-
fault group and priority is global and 10, respectively. For
example,line 2 in Figure14 shows a rule beingenteredinto
thebag large itemsrule group.

Internally, therule

Id in Group at Priority if Condition then Action

is convertedinto two Prologpredicates

<�=�>�?A@�B�C8D�E�FHG�BJI3C"B�ILK"M�F4N�OPFRQ�S�TJC8B�M�UWVYX�Z�C8[�OJILKJI3C8[

and B�=J>Y\�?A@�B�C8D�E�F�N�OPFHQ�S�T�C"B�M]U^V�X`_Ja�K�I3C3[

(seeline 14 in Figure15 andline 16 in Figure15).This sep-
arationpermits the extensive customizationof the forward
chainersincerule conditionscanbetestedwithout triggering
therule action.

ThevariablesGroup,Priority,Id,Memory areused
by PIKE’s conflict resolutionstrategies. Recall that conflict
resolutionis the processingof selectingone rule from the
spaceof rulesof satisfiedconditions.PIKE employs thefol-
lowing conflict resolutionstrategies:

Rulegroups: PIKEmaintainsapointerto somecurrentgroup
in the group/1 fact (seeline 35 in Figure 16). Only
ruleswithin thecurrentgrouparetested.

Priority ordering: Prior to forward chaining,PIKE gathers
togethera list of all theuniquegroupnamesandrule pri-
orities within eachgroup (seeline 20 in Figure 16). At
runtime,rulesareexploredwithin a groupin priority or-
deringfrom priority oneto lowerpriorities(seeline 28 in
Figure16 andline 34 in Figure16).

Refraction: PIKE never fires the samerule actiontwice on
thesamesetof variablebindings.TheMemory argument
of lhs/4 andrhs1/3 containsall thevariablespassed

spec.pl

spec(Helper=Spec,
[ ’:spec’(Helper,F,Term1,Ids,Indicies,Names)
, (:- index(Index))
, (:- dynamic F/Arity)
, (portray(Term1) :- write(F/Arity))
, (touch(Touched,Com1,Final):- Toucher)
, (gets(Helper,Term1,Ids) :- Term1)
, (sets(Helper,Term2,Ids) :- bretract(Term1),

bassert(Term2) )
, (makes(Helper,Term1,Ids) :- bassert(Term1))
, (zaps(Helper,Term1,Ids) :- bretract(Term1))
, (goal_expansion(H3,Body) :- singleton(H3), % : 12

clause(H3,Body))
, goal_expansion(H4,H5a)
, (goal_expansion(H5,true) :- one(H5)) % : 15
, (H1 :- H3)
, Self
| Rest
]) :-

Spec =.. [F|Fields],
length(Fields,Arity),
makeIndex(Fields,1,Indicies,Ids,Args,Names),
H1 =.. [Helper,Com],
H3 =.. [Helper,Com,_,_],

% H3a =.. [Helper,Com,Initial,Final],
H4 =.. [Helper,Field,Old ,In,In],
H5a =.. [Helper,Field,Old,Old,In,In],
H5 =.. [Helper,_,_,_,_,_],
functor(Touched,F,Arity),
Toucher=..[Helper,Com1,Touched,Final],
Index =.. [F|Indicies],
Term1 =.. [F|Args],
Self =.. [Helper,self,In,Out,In,Out],
copy_term(Term1/Ids,Term2/Ids),
findall(One,spec1(Helper,Names,F,Arity,One),Rest).

spec1(Helper,Names,F,Arity,One) :-
nth1(Pos,Names,Item),
joinArgs(F,Arity,Pos,Old,New,T1,T2),
One =.. [Helper,Item,Old,New,T1,T2].

joinArgs(F,Arity,Pos,Old,New,Term1,Term2) :-
length(L1,Arity),
Pos0 is Pos - 1,
length(Before,Pos0),
append(Before,[Old|After],L1),
append(Before,[New|After],L2),
Term1 =.. [F|L1],
Term2 =.. [F|L2].

makeIndex([],_,[],[],[],[]).
makeIndex([+H|L],N,[1|Pos],[Arg|Ids],[Arg|Args] ,[H|T]) :-

N1 is N + 1,
makeIndex(L,N1,Pos,Ids,Args,T).

makeIndex([H|L],N,[0|Pos],Ids,[_|Args],[H|T]) :-
atomic(H),
N1 is N + 1,
makeIndex(L,N1,Pos,Ids,Args,T).

blank(H,B,Id) :- ’:spec’(H,_,B,Id,_,_).

Fig. 12 SeeFigure11 for sampleusage.

fromtheCondition to theAction . Thesesharedvari-
ablesarefoundvia sharedVars/3 shown in Figure17
which is calledat line 29 in Figure15.

Recency:WhenPIKE assertsanything, it is assertedabove
all olderassertions(e.g.seeline 41in Figure10andline 45
in Figure10).Hence,ruleswill fire moreonnewerasser-
tionsthanolderassertions.

ecg.pl

ecg((H*X0 --> Y0),Out) :-
ecg1(Y0,H,Y,W0,W),
X =.. [H,X0,W0,W],
expand_term((X :- Y),Temp),
tidy(Temp,Out).

ecg1(X,H,Y,W0,W) :- once(ecg0(X,Z)), ecg2(Z,H,Y,W0,W).

ecg0(X , leaf(X)) :- var(X).
ecg0(! , !).
ecg0((X -> Y) , two((->),X,Y)).
ecg0((X and Y) , two((, ),X,Y)).
ecg0((X , Y) , two((, ),X,Y)).
ecg0((X or Y) , two((; ),X,Y)).
ecg0((X ; Y) , two((; ),X,Y)).
ecg0(not X , one((not),X)).
ecg0(* Call0 , local(Call)) :- c2l(Call0,Call).
ecg0(H*Call0 , foriegn(Call,H)) :- c2l(Call0,Call).
ecg0(‘X takes Y , ‘X takes Y ).

% : . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 21
ecg0(change X=Id with Y0, with(X,Id,Y,gets,sets)):- w2c(Y0,Y).
ecg0(make X with Y0,with(X,_,Y,blank,makes)):- w2c(Y0,Y).
ecg0(zap X=Id with Y0, with(X,Id,Y,gets,zaps)):- w2c(Y0,Y).
ecg0(zap X=Id, with(X,Id,true,gets,zaps)).
ecg0(X=Id with Y0, with(X,Id,Y,gets,noop)):- w2c(Y0,Y).
ecg0(X with Y0, with(X,_ ,Y,gets,noop)):- w2c(Y0,Y).
% : . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 28

ecg0(X, wrap(X)).

ecg2(!, _,!) --> [].
ecg2(‘List takes Item,H,X) --> ecg1(!List=[Item|‘List],H,X).
ecg2(one(O,A0), H,X) --> ecg1(A0,H,A), X=..[O,A].
ecg2(two(O,A0,B0),H,X) --> ecg1(A0,H,A),ecg1(B0,H,B),

X =.. [O,A,B] .
ecg2(leaf(X), _,X) --> [].
ecg2(wrap(X0), H,X,W0,W) :- wrapper(X0,H,X1),

ecg3(X1,X,W0,W).
ecg2(local(L), H,X) --> calls(L,H,X).
ecg2(foriegn(L,H ),_,X,W, W) :- calls(L,H,X,_,_).
ecg2(with(H,Id,X0,Pre,Post),_,(One,Two,Three),W1,W1 ) :-

One =..[Pre, H,W0,Id],
Three =..[Post,H,W,Id],
ecg1(X0,H,Two,W0,W).

noop(_,_,_).

ecg3([X0],X) --> add2(X0,X).
ecg3([X0,Y|Z],(X,Rest)) --> add2(X0,X), ecg3([Y|Z],Rest).

add2(call(X),X,W,W) :- !.
add2(T0,T,W0,W) :- T0 =.. L0, append(L0,[W0,W],L1), T =.. L1.

calls([Call0],H,Call,W0,W) :- Call =.. [H,Call0,W0,W].
calls([Call0,Call1|Calls0],H,(Call,Calls),W0,W) :-

Call =.. [H,Call0,W0,W1],
calls([Call1|Calls0],H,Calls,W1,W).

w2c(A with B,(A,C)) :- !,w2c(B,C).
w2c(A,A).

Fig. 13

3.4 TheWith Statement

Anotherimportantidiom within PIKE is

Class=Id with Method1 with Method2 with ...

PIKE expandstheseto multiple methodcalls invoked over
thesameobject.Importantvariantsof this idiom are:

– change Class=Id with Method1 with ...
TheMethodsareprefixedby amatchto theobjectandare
followedby anupdateto theobject.

– make Class with Method1 with ...
The Methods arerun on a new objectof the specified

largeitems.pl

bottles
in bag_large_items % : . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 2
if order=I with ‘items has N and

grocery with ‘name=N
with ‘size=large
with *type(bottle) and

bag = B with *largeItems(L) and
L < 6

then change order=I with delete(‘items,N,!items) and
change bag = B with ‘contents takes N

since ’there’’s room in bag ’ and B and ’ for a large bottle’.

largeitems
in bag_large_items
if order=I with ‘items has N and

grocery with ‘name=N
with ‘size=large and

bag = B with *largeItems(L) and
L < 6

then change order=I with delete(‘items,N,!items) and
change bag = B with ‘contents takes N

since ’there’’s room in bag ’ and B and ’ for one ’ and N.

newbag
in bag_large_items
if order with ‘items has N and

grocery with ‘name=N
with ‘size=large

then make bag with *nothing
since ’need a new bag’.

endlarge
in bag_large_items
if true
then goto bag_medium_items
since ’all done with large items’.

Fig. 14

Class andarefollowedby anassertionof theresulting
object.

– zap Class=Id with Method1 with ...
TheMethods areprefixedby a matchto theobjectand
arefollowedby deletionof theobject.TheMethod s all
calledprior to deletion.

– zap Class=Id
TheMethods areprefixedby a matchto theobjectand
arefollowedby deletionof theobject.

The prefix and following codeis addedbetweenline 21 in
Figure13 andline 28 in Figure13

4 Optimizations

PIKE also includesthree optimization methods:unfolding
anda unificationof thematch/selectprocess:

Unfolding: If a clausesub-goalmatchesto a single other
clause,thenthe sub-goalis replacedby the body of the
other clause;seeline 12 in Figure 12. Sometimes,this
unfolding unwrapsto just a true sub-goal;seeline 15
in Figure12.Suchtrue sub-goalsareredundantandare
purgedvia thecodeof Figure18.

Unifiedmatch/select: Theorderin whichPIKE’srulesareto
betestedcanbedeterminedvia therule priority number.
If rules are testedwithin eachgroup in this order, then
thefirst rulewith a satisfiedconditionwouldbethehigh-
estpriority satisfiedrule.By exploringrulesin thisorder,

7



rules.pl

r=rule(group,id,wme,priority).

rules(In,Out) :- once(r(prep(In,Out),_,_)).

r*init --> !id=0,!group=global,!priority=10.

r*head(Id in G at P) --> !id = Id, !group=G, !priority=P.
r*head(Id at P in G) --> !id = Id, !group=G, !priority=P.
r*head(Id in G) --> !id = Id, !group=G.
r*head(Id at P) --> !id = Id, !priority=P.
r*head(Id ) --> !id=Id.

r*prep(X if Y0 then Z0 since Why0,
[(lhs(G,P,Id,Mem) :- % : . . . . . . . . . . . . . . . . 14

Y)
,(rhs1(G,Id,Mem) :- % : . . . . . . . . . . . . . . . . 16

Z,
say(G,Id,Why))

]) --> !,
nl,print(X), write(’? ’),
* init,!,
call(ecgHack(Y0,Y)),
call(ecgHack(Z0,Z)),
call(c2l(Why0,Why)),
* head(X),
‘group=G,
‘id = Id,
‘priority=P,
call(sharedVars(Y,Z,Mem)), % : . . . . . . . . . . . . . . . . 29
write(’ YES!’).

r*prep(X if Y0 then Z0,Out) -->
*prep(X if Y0 then Z0 since [],Out).

ecgHack(X0,X) :-
ecg1(X0,_,X1,_,_),
expand_term(X1,X2),
tidy(X2,X).

rshow(Group,Id) :-
clause(lhs(Group,P,Id,Mem),LHS),
clause(rhs1(Group,Id,Mem),RHS),
portray_clause((lhs(Group,P,Id,Mem) :- LHS)),
portray_clause((rhs1(Group,Id,Mem) :- RHS)).

Fig. 15

PIKE avoids a computationallyexpensive MATCH pro-
cess.;seeline 28 in Figure16 andline 34 in Figure16.

5 A SampleApplication

This papercontainsfull sourcecodefor a PIKE rule/object
systemthat implement’s Patrick Winston’s BAGGERprob-
lem[23]. BAGGERis Winston’sallegory for XCON: XCON
configurescomputersby checkingthe right componentsare
combinedtogetherwhile BAGGERchecksthattheright gro-
ceryordersarecombinedtogetherin grocerybags.

PIKE’sBAGGERis loadedin Figure19.Thesystemcon-
tainsfiverule groups:

Global: Createsa sampleorder;seeFigure20. The cur-
rentgroupis thenchangedto...

Check order: Checksif any itemsaremissingfrom the
order;seeFigure20. The currentgroupis thenchanged
to...

Bag large items: Handlesthe baggingof the bulky
items;seeFigure14. The currentgroupis thenchanged
to...

fchain.pl

refraction=alreadyUsed(+group,+id,mem).
refraction*active.

fchain :-
no(silent),
reset(X),
run(X).

reset(Info) :-
bagof(G/Ps,priorities(G,Ps),Info),
goto global,
forall(active(A),retractall(A)).

active(A) :-
blank(_,A,_), touch(A,active,_).

groups(All) :-
setof(One,AˆBˆCˆDˆclause(lhs(One,A,B,C),D),All).

priorities(Group,All) :- % : . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 20
groups(Groups),
member(Group,Groups),
setof(One,GroupˆBˆCˆDˆclause(lhs(Group,One,B,C),D) ,All).

run(Info) :- step(Info),!, run(Info). run(_).

step(Info) :-
todo(Info,Group,Priority), % : . . . . . . . . . . . . . . . . . . . . 28
lhs(Group,Priority,Id,Mem),
not alreadyUsed(Group,Id,Mem),
assert(alreadyUsed(Group,Id,Mem)),
rhs(Group,Id,Mem).

todo(Info,Group,Priority) :- % : . . . . . . . . . . . . . . . . . . . . 34
group(Group), % : . . . . . . . . . . . . . . . . . . . . 35
member(Group/Orders,Info),
member(Priority,Orders).

rhs(Group,Id,Mem) :- rhs1(Group,Id,Mem),!.
rhs(Group,Id,_) :-

format(’% ?? failed rule action ˜w in ˜w’,[Id,Group]),nl.

Fig. 16

sharedvars.pl

sharedVars(T1,T2,V) :- vars(T1,V1), vars(T2,V2),
sharedVars1(V1,V2,V),!.

sharedVars(_,_,[]).

vars(Term,All) :- setof(One,vars1(Term,One),All).

vars1(Term,V) :- subterm(Term,V), var(V).

sharedVars1([],_,[]).
sharedVars1([H|T0],L,[H|T]) :- member(X,L),H == X,!,

sharedVars1(T0,L,T).
sharedVars1([_|T0],L, T ) :- sharedVars1(T0,L,T).

Fig. 17

Bag medium items: Handlesthe baggingof the mid-
sizeditems;seeFigure21.Thecurrentgroupis thenchanged
to...

Bag small items: Tries to sneakthe small itemsinto
thebagscreatedabove;seeFigure1.

Figure22 showswhathappenswhenthewholesystemis
loadedandexecuted.GiventheORDER
b c B]S�d�OJFReY<�C8E�FHe�B]d8[YC]<�dYFRe�B�d3[YC�<�dPF(I�a8S8Z�B]S�dLTWF
EYC8K]d8K�C�Z3=fI�EJ>Lg

PIKE’sBAGGERgeneratestwo bags:

bagDB(1, [glop, potatoChips, iceCream, bread]).
bagDB(0, [granola, pepsi]).

tidy.pl

tidy(A,C) :-
option(brave)
-> once(tidy1(A,B)),once(tidy1(B,C))
; once(tidy1(A,C)).

tidy1(A, A) :- var(A).
tidy1(X=X, true) :- option(brave).
tidy1(X is Y, true) :- option(brave), ground(Y), X is Y.
tidy1((A :- true), A).
tidy1((A :- B), R) :- tidy(B,TB),

(TB=true -> R=A; R=(A :- TB)).
tidy1((A,B), (A,TB)) :- var(A), tidy(B,TB).
tidy1((A,B), (TA,B)) :- var(B), tidy(A,TA).
tidy1(((A,B),C), R) :- tidy((A,B,C), R).
tidy1((true,A), R) :- tidy(A,R).
tidy1((A,true), R) :- tidy(A,R).
tidy1((A,B), R) :- tidy(A,TA), tidy(B,TB),

(TB=true -> R=TA ; R=(TA,TB)).
tidy1((A;B), (TA;TB)) :- tidy(A,TA), tidy(B,TB).
tidy1((A->B), (TA->TB)) :- tidy(A,TA), tidy(B,TB).
tidy1(not(A), not(TA)) :- tidy(A,TA).
tidy1(A, A).

Fig. 18 Remove redundanttrues.

ruleseg.pl

:- [starlog]. % see Figure 8

:- [grocery % see Figure 2
,order % see Figure 3
,bag % see Figure 4
].

:- [checkrules % see Figure 20
,largeitems % see Figure 14
,mediumitems % see Figure 21
,smallitems % see Figure 1
].

startup
in global
if true
then make bag with *nothing and

make order
with ‘id=1
with ‘items= [bread,glop, granola,

granola,iceCream,
potatoChips] and

goto check_order
since ’BAGGER v3.0 is up and running!!’.

ruleseg :- time(fchain), listing(orderDB), listing(bagDB).

:- demos(ruleseg).

Fig. 19

checkrules.pl

b1
in check_order
if order=I with ‘items has potatoChips and

not (order=I with ‘items has N and
grocery with ‘name=N with *type(bottle)
)

then change order=I with ‘items takes pepsi
since ’order ’ and I and ’ has chips, but needs pepsi’.

b2
in check_order
if true
then goto bag_large_items
since ’all done with checking orders’.

Fig. 20

mediumitems.pl

b8
in bag_medium_items
if order=I with ‘items has N and

grocery with ‘name=N with ‘size=medium and
(bag=B with *empty or

(bag=B with ‘contents has C and
grocery with ‘name=C with ‘size=medium)

)
then change order=I with delete(‘items,N,!items) and

change bag=B with ‘contents takes N
since ’bag ’ and B and ’ can hold item ’ and N.

newbag4medium
in bag_medium_items
if order with ‘items has N and

grocery with ‘name=N
with ‘size=medium

then make bag with *nothing
since ’need a new bag’.

endmedium
in bag_medium_items
if true
then goto bag_small_items
since ’all done with small items’.

Fig. 21

ruleseg.out

% output from ruleseg.pl

[global::startup]
BAGGERv3.0 is up and running!!

[check_order::b1]
order [1] has chips, but needs pepsi

[check_order::b2]
all done with checking orders

[bag_large_items::bottles]
there’s room in bag [0] for a large bottle

[bag_large_items::largeitems]
there’s room in bag [0] for one granola

[bag_large_items::endlarge]
all done with large items

[bag_medium_items::newbag4medium]
need a new bag

[bag_medium_items::b8]
bag [1] can hold item bread

[bag_medium_items::b8]
bag [1] can hold item iceCream

[bag_medium_items::b8]
bag [1] can hold item potatoChips

[bag_medium_items::endmedium]
all done with small items

[bag_small_items::b11]
best to avoid bottles and small items

:- dynamic orderDB/2.

orderDB(1, []).

:- dynamic bagDB/2.

bagDB(1, [glop, potatoChips, iceCream, bread]).
bagDB(0, [granola, pepsi]).

Fig. 22

6 The Restof STARLOG

Therestof STARLOG is containedin thefollowing files:

– Figure23 is thefirst file loadedwhich setscertainglobal
flags.

– Figure24 showsa sub-routineto Figure13.
– Figure25definessomeof theruleconditionsandactions.

flags.pl

:- Stuff=(gets/3, sets/3, makes/3, zaps/3,
’:spec’/6,
lhs/4, rhs1/3,touch/3),

multifile(Stuff),
discontiguous(Stuff),
dynamic(Stuff).

:- index(gets( 1,1,0)).
:- index(sets( 1,1,0)).
:- index(makes(1,1,0)).
:- index(zaps( 1,1,0)).
:- index(lhs(1,1,1,0)).
:- index(rhs1(1, 1,0)).

:- dynamic group/1,
option/1.

yes(X) :- option(X) -> true; assert(option(X)).
no(X) :- retractall(option(X)).

:- yes(brave). % compile time evaluation
:- yes(loadSlowly). % never skippled unchanged files on load
:- no(silent). % don’t suppress rule ’since’ text
:- yes(nervous). % check that fields, methods exist
:- yes(unfold). % replace sub-goals by true

Fig. 23

wrapper.pl

wrapper(X,F,Out) :-
wrap(X,F,Before,[],After,[],Goal),
append(Before,[call(Goal)|After],Out).

wrap(X,F,B0,B,A0,A,Y) :-
once(wrap0(X,Z)),
wrap1(Z,F,B0,B,A0,A,Y).

wrap0(X, leaf(X) ) :- var(X).
wrap0(X, leaf(X) ) :- atomic(X).
wrap0([], leaf(true) ).
wrap0([H|T], [H|T] ).
wrap0(‘X, ‘X ).
wrap0(!X, !X ).
wrap0(X, term(X) ).

wrap1(leaf(X), _,B, B, A, A, X).
wrap1([H0|T0], F,B0,B, A0,A, [H|T]) :-

wrap(H0, F,B0,B1,A0,A1,H),
wrap(T0, F,B1,B, A1,A, T).

wrap1(term(X), F,B0,B, A0,A, Y) :-
X =.. L0,
wrap(L0,F,B0,B,A0,A,L),
Y =.. L.

wrap1(‘X, F,[H|B],B,A,A,Y) :- H=..[F,X,Y,Y].
wrap1(!X, F,B,B,[H|A],A,Y) :- H=..[F,X,_,Y].

Fig. 24

– Figure26 is a file which, if loaded,will exercisemostof
STARLOG’s PIKE.

– Figure27 containscertainProloghackssuchasrepairof
over-zealousDCGexpansion.

– Figure28 testsif only oneclausematchesasub-goal.
– Figure29 is a tool for printing listsof clauses.
– Figure 30 is a tool for running a goal and trapping its

outputto afile.
– Figure31 containsmiscellaneouscode.

verbs.pl

goto Group :- retractall(group(_)), assert(group(Group)).

List has Item :- member(Item,List).

say(_,_,_) :- option(silent),!.
say(_,_,[]) :- !.
say(Group,Id,Words) :- !,

format(’[˜w::˜w] ’,[Group,Id]),nl,
write(’ ’),
forall(member(One,Words),write(One)),
nl.

Fig. 25

egs.pl

:- [speceg].
:- [ecgeg].
:- [ruleseg].

Fig. 26

hacks.pl

%:- yes(brave).

goal_expansion(append(A,B,C,D,D), append(A,B,C)).
goal_expansion(once(A,B,B), once(A)).
goal_expansion(=..(A,B,C,C), =..(A,B)).
goal_expansion(=(A,B,C,C), =(A,B)).
goal_expansion(call(A,B,B), A).
%goal_expansion(call(A), A).
goal_expansion(noop(_,_,_), true).
%goal_expansion(X=X, true) :- option(brave).

prolog_listing:put_tabs(N) :-
N > 0, !,
write(’ ’),
NN is N - 1,
prolog_listing:put_tabs(NN).

prolog_listing:put_tabs(_).

Fig. 27

singleton.pl

singleton(X) :-
bagof(X/Y,clause(X,Y),[_]).

one(X) :- singleton(X),X.

Fig. 28

show.pl

show(X) :- show(X,_).

show([],_).
show([H|T],X) :-

once(show1(H,Y,S)),
(X=Y -> true; nl),
show2(S),
show(T,Y).

show1((X :- true),F/A, X) :- functor(X,F,A).
show1((X :- Y), F/A,(X:-Y)) :- functor(X,F,A).
show1(X, F/A, X) :- functor(X,F,A).

show2((X :- Y)) :- !,portray_clause((X:-Y)).
show2(X) :- numbervars(X,1,_), print(X), write(’.’), nl.

Fig. 29 A simpleprettyprint.

demos.pl

demos(G) :-
sformat(Out,’˜w.out’,G),
(exists_file(Out) -> delete_file(Out) ; true),
write(Out),nl,nl,
tell(Out),
format(’% output from ˜w.pl’,G),nl, nl,
T1 is cputime,
ignore(forall(G,true)),
T2 is (cputime - T1)*1000,
nl,format(’% runtime = ˜w sec(s)’,[T2]),nl,
told,
format(’% output from ˜w.pl’,G),nl,
ignore(forall(G,true)),
nl,format(’% runtime = ˜w sec(s)’,[T2]).

Fig. 30 Runa goal,trap it’ s outputto file and,also,show it on the
screen.

misc.pl

l2c([X,Y|Z],(X,Rest)) :- !, l2c([Y|Z],Rest).
l2c([X],X).

c2l(X,[X]) :- var(X),!.
c2l((X and Y),[X|Rest]) :- !, c2l(Y,Rest).

c2l((X,Y),[X|Rest]) :- !, c2l(Y,Rest).
c2l(X,[X]).

subterm(X,X).
subterm(In, X) :- compound(In), arg(_,In,Arg), subterm(Arg,X).

term2list(Term0, L) :-
Term0 =..L0,
once(maplist(term2list1, L0, L)).

term2list1(H,H) :- var(H).
term2list1(H,H) :- atomic(H).
term2list1(H0,H) :- term2list(H0,H).

ensure(X) :- X,!.
ensure(X) :- bassert(X).

bassert(C) :- asserta(C).
bassert(C) :- retract(C),!, fail.

bretract(C) :- retract(C) , bretract1(C).

bretract1(_).
bretract1(C) :- asserta(C), fail.

Fig. 31

7 Conclusion

Purerule-basedprogramminghad many proponentsin the
early daysof knowledgeengineering.Theseproponentsbe-
camefewer in numberas more and more developerswere
forcedto extendpurerule-basedprogramming.

We have arguedherethat suchextensionsarenecessary
and simple.The Prolog codeshown hereis very brief and
implementsa object/ruleinterpreter/optimizer. Sucha small
systemiseasilycustomizedandsupportsourpreferredtoolkit
for knowledgeengineering:theencodingof domain-specific
knowledgecliches.

